

---

ANALYZ (COEF=*input parameter vector*, CONSTR, NAMES=(*list of names*),  
PRINT, SILENT, TERSE, VCOV=*matrixname*)  
*list of equation names* ;

---

**Function:**

ANALYZ computes the values and estimated covariance matrix for a set of (nonlinear) functions of the parameters estimated by the most recent OLSQ, LIML, LSQ, FIML, PROBIT, etc. procedure. It also computes the Wald test for the hypothesis that the set of functions are jointly zero. If the functions are linear, after an OLSQ command, the F test of the restrictions, and implied restricted original coefficients will be printed.

The method used linearizes the nonlinear functions around the estimated parameter values and then uses the standard formulas for the variance and covariance of linear functions of random variables. See the references for further discussion of this "delta method". TSP obtains analytic derivatives internally for the nonlinear functions. ANALYZ can also be used to select/reorder a subset of a VCOV matrix and COEF vector, for use in making a Hausman specification test.

**Usage:**

ANALYZ is followed by a list of equation (FRML) names. After estimation procedures with linear models (OLSQ, INST, LIML, PROBIT, ...), these equations specify functions of the estimated coefficients which are to be computed by referring to the coefficients by the names of the associated variables. After estimation procedures with nonlinear models (LSQ and FIML), the equations specify functions of the estimated parameters. ANALYZ has no provision for combining the variances from more than one estimation, since it cannot obtain the associated covariance. The equations must be previously defined by FRML statements; if the FRML statements have variable names on the left hand side, the computed value of each function will be stored under that variable name.

If series names (other than the names of right hand side variables from the previous OLSQ, INST, LIML, or PROBIT estimation) are included in the FRML(s), they must evaluate to scalars, either through a subscript, or with a SMPL that is one observation long (such as SMPL 10,10;). One application for this kind of FRML is an elasticity which depends on estimated parameters, and also on data such as income. ANALYZ will compute the standard error for such a FRML using the covariance matrix of the estimated parameters, and treating the data as fixed constants. See the last example below.

**Options:**

**COEF**= vector containing the values of the parameters in the equations to be analyzed. This vector should correspond to the parameters listed in the NAMES= option, and also to the supplied VCOV matrix. The default is @COEF.

**CONSTR/NOCONSTR** allows calculation of constrained regression coefficients, when ANALYZ follows OLSQ. Use NOCONS to prevent the constrained coefficients from being calculated or printed.

**NAMES**= specifies an optional list of parameter names which are the labels for an associated covariance matrix supplied by the VCOV= option. The default is @RNMS.

**PRINT/NOPRINT** tells whether or not the ANALYZ inputs (parameter values and equations) are to be printed. Under the default, NOPRINT, only the results are printed.

## ANALYZ

**SILENT/NOSILENT** specifies that no output is to be produced. The results are stored under the names @RNMSA, @COEFA, etc.

**TERSE/NOTERSE** specifies that "minimal" output is to be produced. When ANALYZ follows OLSQ, this suppresses the chi-squared test, but still prints the F test. It also suppresses printing of the constrained coefficients, although they are still stored (under @COEFC, etc.), unless NOCONST has been used.

**VCOV=** specifies the name of a variance-covariance matrix of the input parameters (whose names are given by NAMES=). The use of these two options enables one to do an ANALYZ on matrices other than the @VCOV matrix from a standard estimation procedure. The default is @VCOV.

### Examples:

Obtain "long-run" coefficients for models with lagged dependent variables:

```
FRML LR1 ALPHALR = ALPHA/(1-LAMBDA) ;
FRML LR2 PHILR = PHI/(1-PSI) ;
ANALYZ LR1,LR2 ;
```

See the **EQSUB** command in this **Reference Manual** for an example of using ANALYZ (with EQSUB) to evaluate and obtain standard errors for restricted parameters in a translog system.

The next example shows two ways to calculate an elasticity (and its standard errors) for both a given observation in the data, and at the mean of the sample data:

```
FRML EQ1 LQ1 = A1 + B1*LP1 + B2*LP2 + B12*LP1*LP2 + B13*LP1*LP3 + B23*LP2*LP3;
FRML EL1 ELD1 = B1 + B12*LP2 + B13*LP3;           ? d(LQ1)/d(LP1)
SMPL 48,95;
LSQ EQ1;
SMPL 72,72;                                     ? Obtain elasticity at 1972 values of LP2, LP3, etc.
ANALYZ EL1;
SMPL 48,95;                                     ? To evaluate elasticity at sample mean values of LP2, LP3, etc.
MSD LP2 LP3;                                    ? create an artificial '1996' observation which holds the means.
SMPL 96,96;
UNMAKE @MEAN LP2 LP3;
ANALYZ EL1;
```

Here is an example of using ANALYZ after OLSQ. It computes a chi-squared test of the hypothesis that the sum of the two coefficients is zero (this test statistic equals the standard F-statistic).

```
OLSQ Y C X1 X2 ;
FRML SUM X1+X2 ;
ANALYZ SUM ;
```

Suppose that we want to extract a few parameters and their associated VCOV matrix from a system with a large number of parameters in arbitrary order:

```
SUPRES COEF;
LSQ(SILENT) EQ1-EQ50;                            ? estimation with a large number of equations
SUPRES;
DOT B1-B5;
FRML EQ. . = . ;                                  ? construct FRML EQB1 B1 = B1; etc.
ENDDOT;
ANALYZ EQB1-EQB5;                                ? print results for 5 of the parameters only
RENAME @VCOVA VCVB1_5;                            ? for use later
```

RENAME @COEFA CB1\_B5;

### Output:

If the PRINT option is on, ANALYZ prints a title, the names of the input parameters, the equations in symbolic form, a table of the derived functions and their standard errors, and the chi-squared value of a test that the functions are jointly zero. This chi-squared has degrees of freedom equal to the number of equations. The P-value (significance level) for the chi-squared test is also printed. If the print option is off (the default), only the derived functions and the chi-squared test are printed.

ANALYZ also stores the calculated parameters and their variances in data storage as though they were estimation results, whether or not the PRINT option is on. The results are stored under the following names:

Name	Type	Length	Variable Description
@RNMSA	list	#eqs	Names of derived parameters.
@WALD	scalar	1	Value of Wald test.
@NCOEFA	scalar	1	Number of derived parameters.
@NCIDA	scalar	1	Degrees of freedom.
%WALD	scalar	1	P-value (significance) of Wald test.
@COEFA	vector	#eqs	Values of derived parameters.
@SESA	vector	#eqs	Standard errors of derived parameter.
@TA	vector	#eqs	T-statistics (asymptotically normal).
@VCOVA	matrix	#eqs*#eqs	Estimated variance covariance of derived parameters.

### Method:

Assume that a previous estimation in TSP has stored a vector of K parameter estimates  $b = @COEF$  and their variance covariance matrix  $var(b) = @VCOV$ , and values and standard errors for the M functions  $f(b)$  are desired. ANALYZ obtains the first derivatives of  $f$  with respect to  $b$  analytically:  $df/db = G$  (a  $K \times M$  matrix). The functions  $f(b)$  and the matrix  $G$  are evaluated at the current values of  $b$ , along with any other constants or data values which may appear in  $f(b)$ . The variance-covariance matrix for  $f(b)$  is then (asymptotically, or exactly if  $f(b)$  is linear) defined as  $var(f(b)) = G'@VCOV * G$ . This is known as the "delta method". For example, if  $M=1$  and  $f(b) = f1 = 2*b1$ , then  $G=2$  (with zeros elsewhere if  $K>1$ ), and  $var(f1) = 4*var(b1)$ .

If the equations are linear, and an OLSQ command was used for estimation, ANALYZ prints the F-statistic for the set of joint restrictions ( $@FST = @WALD/@NCIDA$ ). In addition, ANALYZ computes and prints the implied restricted original coefficients and their standard errors. These are stored under  $@COEFC$ ,  $@VCOVC$ , etc.

### References:

Bishop, Y.M.M., S.E. Fienberg, and P.W. Holland, **Discrete Multivariate Analysis: Theory and Practice**, MIT Press, Cambridge, MA, 1975, pp. 486-502.

Gallant, A. Ronald, and Dale Jorgenson, "Statistical Inference for a System of Simultaneous, Non-linear, Implicit Equations in the Context of Instrumental Variable Estimation", **Journal of Econometrics** **11**, 1979, pp. 275-302.

Gallant, A. Ronald, and Alberto Holly, "Statistical Inference in an Implicit, Nonlinear, Simultaneous Equation Model in the Context of Maximum Likelihood Estimation", **Econometrica** **48**, 1980, pp. 697-720.